/*******************************************************************************
*   			Impacts of a Large-Scale Parenting Program: 				   *
*					Experimental Evidence from Chile						   *
********************************************************************************


	REQUIRES:	"${dt_final}panel_RR_indexes.dta"
	CREATES:	${out_tables}/T4_ITT_parentalpractices.xls
				
	WRITEN BY:  Italo Lopez  [italolop@usc.edu]

********************************************************************************
	Prepare data
*******************************************************************************/

	* Load data
	use "${dt_final}panel_RR_indexes.dta", clear


	* Establish macros for regressions

/*******************************************************************************
*   Set macros for regressions
*******************************************************************************/
	
	* Outcomes
	*-----------------------------------------------------------
	local childoutcomes  "zDCCSPunt zptevir_irtscore2 zt_score zcbcl_t_e zcbcl_t_i zibatt_sspi zibatt_ssai zibatt_sssr zas_raw_score" 
	local behaviors "home_score se_score disc_score" 
	local beliefs "zpstyle1 zpstyle2 zpstyle3 zPSCS zPSSS_family zPSSS_friends zPSSS_others zPSI_Distress zCESD zpacotis"	
	
	* Controls
	*-----------------------------------------------------------

	macro  def Xvar1 "i.age_year i.gender"  	
	macro  def Xvar2 "i.age_year i.gender inc_qtaut_old i.pc_edu_mdsfin hh_mem hh_tipo"  	
	macro  def Xvar2old "i.gender inc_qtaut_old i.pc_edu_mdsfin_old hh_mem hh_tipo"  

	
	* Parental characteristics: WAIS and BFI and imputed wais 
	*-----------------------------------------------------------

	macro def parent_cha4 "BFI_ext BFI_ope BFI_con BFI_neu BFI_agr wais2 flag_wais"
	
	
/*******************************************************************************
	Table 1: ITT estimates of parental indexes, all specifications
*******************************************************************************/

	*-----------------------------------------------------------------
	*  Table 4: ITT estimates of parental indexes, all specifications
	*-----------------------------------------------------------------


	local parental "zfbeh_gsem zfbel_gsem zfwb " 
	#delimit;
	foreach var of local parental {;
				
					
				* PANEL 1: ITT with Age and Gender and SES;
					xi: xtreg `var' i.TIPO $Xvar1 if x==1,  fe i(CENTRO_SALUD);
					estimates store `var'_itt1;
					test _ITIPO_2==_ITIPO_3;
					estadd scalar pb=r(p);
					
				* PANEL 2: ITT plus parental characteristics -; 	  
					xi: xtreg `var' i.TIPO $Xvar2 $parent_cha4 if x==1, fe i(CENTRO_SALUD);
					estimates store `var'_itt2;
					test _ITIPO_2==_ITIPO_3;
					estadd scalar pb=r(p);		
				
	};


	#delimit cr
	***  Tables ITT Outcomes

	esttab zfbeh_gsem_itt1 zfbel_gsem_itt1   zfwb_itt1                   /*
		*/ using "${out_tables}/T4_ITT_parentalpractices.csv", replace f		  /* 
		*/ keep(_ITIPO_2 _ITIPO_3) 											  /*
		*/ cells("b(fmt(3)star)" "se(fmt(3)par)") 							  /*
		*/ star(* 0.10 ** 0.05 *** 0.01) 									  /*
		*/ stats(r2 N pb, fmt(2 0 3) 										  /*
		*/ labels("R-squared" "Observations" "P-value NEP-B=NEP-I")) legend	/*
		*/ mtitles("Behavioral Index""Beliefs Index""Well-being Index" ) /*
		*/ title("PANEL 1: ITT with Age and Gender") 	varlabels(_ITIPO_2 "NEP-B" _ITIPO_3 "NEP-I")

		esttab zfbeh_gsem_itt2 zfbel_gsem_itt2   zfwb_itt2        					  /*
		*/ using "${out_tables}/T4_ITT_parentalpractices.csv", append f		  /* 
		*/ keep(_ITIPO_2 _ITIPO_3) 											  /*
		*/ cells("b(fmt(3)star)" "se(fmt(3)par)") 							  /*
		*/ star(* 0.10 ** 0.05 *** 0.01) 									  /*
		*/ stats(r2 N pb, fmt(2 0 3) 										  /*
		*/ labels("R-squared" "Observations" "P-value NEP-B=NEP-I")) legend	/*
		*/ mtitles("Behavioral Index""Beliefs Index""Well-being Index" ) /*
		*/ title("PANEL 2: ITT plus parental characteristics") 	varlabels(_ITIPO_2 "NEP-B" _ITIPO_3 "NEP-I")
		
		
* End of dofile 
